
Chapter 9 

Installing and Configuring 
Server Core 

The usual progression for an operating system (or an application, for that matter) is to 
grow and add features, sometimes well beyond what any of us want or need. Windows 
Server 2008 reverses that trend with a completely new installation option—Server Core. 
When you install Windows Server 2008, regardless of which edition you’re installing, you 
have the option of choosing a full installation, with everything, or just the Server Core 
portion.  

Server Core is just the essentials, with little or no graphical interface. The logon provider 
has the same graphical look, but then, when you’ve logged in, all you see is a single 
command-shell window, as shown in Figure 9-1.  

 
Figure 9-1 The Windows Server 2008 Core desktop. 

Note For improved readability in screen shots used here and in the rest of the book, I’ve 

changed the default color scheme for Command Prompt windows to dark blue text on a 

white background.  

Benefits of a Server Core Installation 
All Windows Server 2008 editions support Server Core, with the exception of Compute 
Cluster Edition. And installing Server Core doesn’t give you a break on the cost of the 
license—it’s exactly the same license and media as the full Windows Server 2008 
installation. At install time, you simply choose which edition you are installing. So, if you 
don’t save any money, and you don’t have special media, and you have reduced 
functionality, why in the world would you choose Server Core over the full product? It’s 
simple, really: security and resources. Let’s take a look at those two in a bit more depth 
before we go on to the details of how to actually install and configure Server Core.  

Security 

In the old days, whenever you installed Windows Server, it automatically installed just 
about everything that was available, and turned on all the services that you were likely to 
need. The goal was to make installation as simple as possible, and this seemed like a good 
idea at the time. Sadly, the world is not a friendly place for computers any more, and that 
approach is no longer safe or wise. The more services that exist, and the more services that 
are enabled, the more attack vectors the bad guys have to work with. To improve security, 
limiting the available attack surfaces is just good common sense.  

In Server Core, Microsoft has completely removed all managed code, and the entire .NET 
Framework. This leaves a whole lot fewer places for possible attack. This does, obviously, 
impose some severe limits on what you can and can’t do with a Server Core installation. 
And it also means that there isn’t any PowerShell possible, which in our opinion is easily 
the biggest limitation of Server Core—but one that we hope will be resolved in a later 
version of Windows Server.  

The default installation of Server Core has only 38 services running.  A typical full Windows 
Server 2008 installation, with one or two roles enabled, is likely to have 60 or even 70 or 
more services running. Not only does the reduced number of services limit the potential 
attack surface that must be protected, but it also limits the number of updates that are 
likely to be required over the life of the server, making it easier to maintain.  

Resources 

The second major benefit to running Server Core is the reduced resources required for the 
base operating system. While the official requirements for installing Windows Server 2008 
are the same for Core as for a full installation, the effective numbers are significantly less, 
in our experience—with the exception of the disk space required (only 2–3 GB of HD 
space for a running Core installation). Plus with the limited subset of tasks that you can 
perform, we think Server Core is ideal for running those infrastructure tasks that everyone 
runs, and that doesn’t require much interaction over time. Tasks such as DHCP, DNS, and 
increasingly Virtualization. Now if it just had PowerShell. 

Installing Server Core 
Installing Windows Server 2008 Server Core is ultimately the same as installing the full 
graphical version of Windows Server 2008. The installation engine is the same, and the 
only difference occurs during the install, when you have to choose which version of 
Windows Server 2008 to install, as shown in Figure 9-2.  

Figure 9-2 During initial installation, you make an irrevocable choice between Server and Server Core. 

Once installation completes, you’re presented with the initial logon screen. Log on as 
Administrator, with no password, and you’ll be immediately prompted to change the 
password and then logged on to the desktop, as shown earlier in Figure 9-1. All initial 
configuration takes place from the command line, though once you’ve configured the 
basics, you’ll be able to use familiar management consoles remotely.  

You can use an unattend.xml file to automate the initial install and configuration of your 
Server Core installation. For details on the settings and syntax of unattend.xml, see 
http://go.microsoft.com/fwlink/?LinkId=81030.  

Configuration 
All configuration tasks for Server Core can be done at the command line, and all the initial 
tasks have to be done at either the command line or as part of the installation process by 
using an unattend.xml script. Once you’re performed these initial configuration tasks, you 
can then use regular Windows management consoles to manage the additional settings. 
Unfortunately, there isn’t a single command shell for the tasks, but a collection of old 
favorites, each with a different behavior and syntax.  

Initial Configuration 
The initial steps you’ll need to perform on a Server Core installation will depend somewhat 
on your intended use of the installation, but we think that the following ones are the most 
obvious: 

• Set a fixed IP address. 

• Change the server name to match your internal standards. 

• Join the server to a domain. 

• Change the default resolution of the console. 

• Enable remote management through Windows Firewall. 

• Enable remote desktop. 

• Activate the server. 

We’ll walk through these steps for you, and leave you with a couple of basic scripts that 
you can modify to automate these tasks for your environment. Table 9-1 contains the 
settings we’ll be using during this install scenario.  

Table 9-1 Settings for Initial Server Core Configuration (Example) 

Setting Value 

IP Address 192.168.51.4 

Gateway 192.168.51.1 

DNS Server 192.168.51.2 

Server Name Hp350-core-04 

Domain To Join example.local 

Default Desktop Resolution 1024x768 

Remote Management Enable for Domain Profile 

Windows Activation Activate  

Set IP Address 

To set the IP address for the server, you need to use the netsh command-line tool. Follow 
these steps to configure TCP/IP: 

1. From the command window, use netsh to get the “name” (index number) of the 
network card. 

netsh interface ipv4 show interfaces 

2. The result will be something like the following:  

C:\Users\administrator>netsh interface ipv4 show interfaces 

 

Idx  Met   MTU   State        Name 

---  ---  -----  -----------  ------------------- 

  2   10   1500  connected    Local Area Connection 

  1   50 4294967295  connected    Loopback Pseudo-Interface 1 

3. The Idx value for your real network card (2, in this case) will be used as the name 
value in future commands for netsh. 

4. Now, using the Idx value from step 2, run the following netsh command: 

netsh interface ipv4 set address name=”<Idx>” source=static address=<IP Address> 
mask=<netmask> gateway=<IP Address of default gateway> 

5. Next, specify the DNS server for the adapter, using netsh again: 

netsh interface ipv4 add dnsserver name=”<Idx>” address=<IP Address of DNS Server> 
index=1 

6. For secondary DNS servers, repeat the command in step 5, increasing the index 
value by one each time.  

Renaming the Server and Joining to a Domain 

The next step in initial configuration is assigning the name of the server and joining it to a 
domain. During initial installation of Windows Server 2008, an automatically generated 
name is assigned to the server and it is placed in the WORKGROUP workgroup. You’ll 
want to change this to align the computer name with your corporate naming policy and 
join the server to the correct domain and Organizational Unit. Our naming policy here has 
three parts: the model of server, the functional role, and a number reflecting its IP address. 
Thus the Server Core computer we’re building in this chapter is named hp350-core-04: it’s 
a Hewlett Packard ML 350 G5 server, it is running Server Core, and the final octet of its IP 
address is four. Your server naming convention will undoubtedly be different, but the 
important thing is to be consistent. Our domain for this book is example.local. 

To change the name of the server and join it to the example.local domain, follow these 
steps: 

1. From the command prompt, use the netdom command to change the name of the 
server:  

netdom renamecomputer %COMPUTERNAME% /newname:<newname> 

2. After you change the name, you must reboot the server.  

shutdown /t 0 /r 

3. After the server restarts, log on to the Administrator account.  

4. Use the netdom command again to join the domain.  

netdom join %COMPUTERNAME% /DOMAIN:<domainname> /userd:<domain admin account> 
/password:* 

5. You’ll be prompted for the password for the domain administrative account you 
used. Enter the password. When the domain join has succeeded, you’ll again need 
to reboot the server.  

shutdown /t 0 /r 

6. After the server restarts, log back on to a domain administrator’s account. (You’ll 
need to click Change User because the server will default to the local administrator 
account.) 

Scripting Initial Configuration 
If you set up more than one or two Server Core computers, you’ll quickly get tired of 
doing all this interactively from the command prompt. We know we did. You have 
the choice of either using an unattend.xml file to set options during the install or 

using simple scripts to automate the process. Both work, and both have their 
adherents, but we tend to use scripts after the fact. You can modify the following 
three scripts (which you’ll also find on the companion CD) for your environment to 
automate the initial TCP/IP, server name, and domain join steps.  

The first script sets the IP address, sets the DNS server, and changes the server name. 

echo off 

REM filename: initsetup1.cmd 

REM 

REM initial setup for a Server 2008 Server Core installation.  

REM command file 1 of 3 

REM 

REM Created: 4 September, 2007 

REM ModHist: 5/9/07 - switched to variables (cpr) 

REM 

REM Copyright 2007 Charlie Russel and Sharon Crawford. All rights reserved. 

REM   You may freely use this script in your own environment, modifying it 

REM   to meet your needs. But you may not re-publish it without permission.  

 

REM first, set a fixed IP address. You'll need to know the index number 

REM of the interface you're setting, but in a default Server Core install,  

REM with only a single NIC, the index should be 2. To find the index,  

REM you can run: 

REM      netsh interface ipv4 show interfaces 

REM  

SETLOCAL 

REM Change the values below to match your needs 

SET IPADD=192.168.51.4 

SET IPMASK=255.255.255.0 

SET IPGW=192.168.51.1 

SET DNS1=192.168.51.2 

SET NEWNAME=hp350-core-04 

 

netsh interface ipv4 set address name="2" source=static address=%IPADD% 
mask=%IPMASK% gateway=%IPGW% 

 

REM Next, set DNS to point to DNS server for example.local. 

REM   192.168.51.2 in this case 

netsh interface ipv4 add dnsserver name="2" address=%DNS1% index=1 

 

REM Now, we need to change the computer name. After we're done, the server  

REM must be restarted, and we can continue with the next batch of commands.  

REM we use the /force command here to avoid prompts 

netdom renamecomputer %COMPUTERNAME% /newname:%NEWNAME% /force 

 

@echo If everything looks OK, the it’s time to reboot 

pause 

REM now, shutdown and reboot. No need to wait.  

shutdown /t 0 /r 

The second script we use is to actually join the server to the domain.  

@echo off 

REM Filename: initsetup2.cmd 

REM 

REM initial setup for a Server 2008 Server Core installation.  

REM command file 2 of 3 

REM 

REM Created: 4 September, 2007 

REM ModHist:  

REM 

REM Copyright 2007 Charlie Russel and Sharon Crawford. All rights reserved. 

REM   You may freely use this script in your own environment, modifying it  

REM   to meet your needs. But you may not re-publish it without permission.  

 

SETLOCAL 

SET DOMAIN=example.local 

SET DOMADMIN=Administrator 

 

REM Join the domain using the netdom join command. Prompts for password 

REM of domain administrator account set above  

 

netdom join %COMPUTERNAME% /DOMAIN:%DOMAIN% /userd:%DOMADMIN% /passwordd:*  

REM now, shutdown and reboot. No need to wait, and that's all we can do  

REM at this time 

 

shutdown /t 0 /r 

Finally, use the third script to enable remote management and activate the server. 

echo off 

REM initsetup3.cmd 

REM 

REM initial setup for a Server 2008 Server Core installation.  

REM command file 3 of 3 

REM 

REM Created: 4 September, 2007 

REM ModHist:  

REM 

REM Copyright 2007 Charlie Russel and Sharon Crawford. All rights reserved. 

REM   You may freely use this script in your own environment, modifying it  

REM   to meet your needs. But you may not re-publish it without permission.  

 

REM Use netsh to enable remote management through the firewall for the  

REM domain profile. This is the minimum to allow using remote MMCs to work 

REM from other computers in the domain.  

 

netsh advfirewall set domainprofile settings remotemanagement enable  

REM allow remote administration group 
netsh advfirewall firewall set rule group="Remote Administration" new enable=yes 

 

REM Allow remote desktop  

REM (also works with group=”Remote Desktop” instead of name=) 

netsh advfirewall firewall set rule name="Remote Desktop (TCP-In)" new enable=yes 

 

REM Enable Remote Desktop for Administration, and allow  

REM downlevel clients to connect 

cscript %windir%\system32\scregedit.wsf /AR 0 

cscript %windir%\system32\scregedit.wsf /CS 0 

 

REM Now, run the activation script 
REM No output means it worked 

Slmgr.vbs -ato 

 

Setting Desktop Display Resolution 

To set the display resolution for the Server Core desktop, you need to manually edit the 
registry. We’d give you a script to do it, but it is dependent on correctly identifying the 
specific GUID for your display adapter. Not something we want to automate. So, to 
change the resolution on your Server Core desktop, follow these steps: 

1. Open regedit. 

2. Navigate to HKLM\System\CurrentControlSet\Control\Video. 

3. One or more GUIDs is listed under Video. Select the one that corresponds to your 
video card. Hint: they each have a device description under the 0000 key that can 
sometimes help. 

4. Under the GUID select the 0000 key, and add a DWORD 
DefaultSettings.XResolution. Edit the value to the X axis resolution you want. For a 
width of 1024 pixels, use 400 hexadecimal, as shown in Figure 9-3.  

Figure 9-3 Editing the display resolution value for the X axis. 

5. Add a DWORD DefaultSettings.YResolution. For  height of 768 pixels, use 300 
hexadecimal.  

Note In some cases, these keys will already exist. If they do, you can simply change their 

value as necessary.  

6. Exit the registry editor and log off using the following: 

shutdown /l 

7. Once you log back on, the new display settings will take effect.  

Enabling Remote Management 

To allow access to the familiar graphical administration tools, you need to enable them to 
work through Windows Firewall. This requires another set of netsh commands. Use the 
following steps to enable remote administration and Remote Desktop: 

1. From the command prompt, use the netsh command to enable remote 
management: 

netsh advfirewall set domainprofile settings remotemanagement enable 

2. Now, enable the Remote Administration group of firewall rules. 

netsh advfirewall firewall set rule group="Remote Administration" new enable=yes 

3. Finally, life is easier when you can connect using remote desktop, so let’s enable 
that, too:  

netsh advfirewall firewall set rule name="Remote Desktop (TCP-In)" new enable=yes 

You should now be able to do additional management using familiar graphical tools from 
another server but connecting to the Server Core computer. 

Activating the Server 

The final step in basic configuration of the Server Core computer is to activate it. This 
requires using a Visual Basic script, which is provided. Use the following command:  

Slmgr.vbs -ato 

Note All the basic initial setup commands for Server Core are included in the three 

scripts described in the UnderTheHood sidebar, and are also available on the CD that 

comes with the book.  

Installing Roles 

Windows Server 2008 Core doesn’t support all the possible roles and features of the full 
graphical Windows Server, but it does support the most important infrastructure roles. We 
think one of the most compelling scenarios for Server Core is as a remote site server to 
enable basic functionality at a remote site where nobody is available on site to administer 
it. By combining the DHCP Server, DNS Server , File Services, and Print Services roles with a 
read-only Active Directory Domain Services role, you have a “branch office in a box” 
solution—just add a remote access device such as a VPN router and you’re in business.  

The File Services role is added by default as part of the base Server Core installation, but 
you can add additional role services to support additional functionality.  

The command used to install a role in Server Core is Ocsetup.exe. The exact same 
command is used to uninstall a role, but with the /uninstall command-line parameter. The 
full syntax for Ocsetup is: 

Ocsetup </?|/h|/help> 

Ocsetup <component> [/uninstall][/passive][/unattendfile:<file>] 
[/quiet][/log:<file>][/norestart][/x:<parameters>] 

The important thing to remember about Ocsetup is that it is quite unforgiving. It is case 
sensitive, and even a slight mistake in the case of the component name will cause the 
command to fail.  

A script to install the roles for this solution, except the domain controller role, would look 
like this:  

@REM filename: SetupBranch.cmd 

@REM 

@REM Setup file to install roles for a branch office server 

@REM 

@REM Created: 5 September, 2007 

@REM ModHist:  

@REM 

@REM Copyright 2007 Charlie Russel and Sharon Crawford. All rights reserved 

@REM   You may freely use this script in your own environment,  

@REM   modifying it to meet your needs.  

@REM   But you may not re-publish it without permission.  

@REM Using "start /w" with ocsetup forces ocsetup to wait until it completes before  

@REM going on to the next task.  

 

@REM Install DNS and DHCP 

@echo Installing DNS and DHCP roles... 

start /w ocsetup DNS-Server-Core-Role 

start /w ocsetup DHCPServerCore 

 

@REM Now, install File Role Services 

@echo Now installing File Role Services... 

start /w ocsetup FRS-Infrastructure 

start /w ocsetup DFSN-Server 

start /w ocsetup DFSR-Infrastructure-ServerEdition 

 

@REM Uncomment these two lines to add NFS support 

@REM start /w ocsetup ServerForNFS-Base 

@REM start /w ocsetup ClientForNFS-Base 

 

@REM Install Print Server Role 

@echo Installing Print Server Role 

start /w ocsetup Printing-ServerCore-Role 

@REM Uncomment next for LPD support 

@REM start /w ocsetup Printing-LPDPrintService 

Note You can’t include the DCPromo command in the script above because installing 

the Print Server role requires a reboot, and locks out DCPromo.  

You cannot use DCPromo interactively to create a domain controller – you must create an 
unattend.txt file to use with it. The basic minimum unattend.txt file is: 

[DCInstall] 

InstallDNS = Yes 

ConfirmGC = yes 

CriticalReplicationOnly = No 

RebootOnCompletion = No 

ReplicationSourceDC = hp350-dc-02.example.local 

ParentDomainDNSName = example.local 

ReplicaOrNewDomain = ReadOnlyReplica 

ReplicaDomainDNSName = example.local 

SiteName=Default-First-Site-Name 

SafeModeAdminPassword = <passwd> 

UserDomain = example 

UserName = Administrator 

Password = <passwd> 

Important The passwords fields must be correct, and will be automatically stripped from 

the file for security reasons. For Server Core, you must specify a ReplicationSourceDC 

value. You should set ReplicaOrNewDomain to the value shown here and ReadOnlyReplica 

to create a read-only domain controller.  

To install the read-only Domain Controller role, follow these steps: 

1. Use Notepad or your favorite ASCII text editor (we use GVim, which works quite well 
in Server Core) to create an unattend.txt file with the necessary settings for the 
domain you will be joining. The specific filename of the unattend file is not 
important because you specify it on the command line. 

2. Change to the directory that contains the unattend file. If the server has any 
pending restarts, you must complete them before promoting the server to domain 
controller. 

3. Run DCPromo with the following syntax:  

Dcpromo /unattend:<unattendfilename> 

4. If there are no errors in the unattend file, DCPromo will proceed and promote the 
server to be a read-only domain controller, as shown in Figure 9-4.  

 
Figure 9-4 Use DCPromo to create a read-only domain controller with an unattend file. 

Listing Roles 

The Oclist.exe command provides a complete list of the available Server Core roles, role 
services, and features, as well as their current state. Use Oclist to get the exact, case-
sensitive list of the features and roles you want to install.  

Managing a Server Core Computer 
Managing a Server Core computer is a different experience for most system 
administrators. None of the graphical tools you’re used to using is available on the server.  
But once you’ve configured the Server Core computer for remote management, as 
described under “Initial Configuration” earlier in the chapter, you can create management 
consoles that point to the Server Core computer, which allow you to do all your tasks from 
a graphical console.   

More Info For details on how to create custom MMCs, see Chapter 14, “Managing 

Daily Operations.” 

There are four basic ways to manage a Server Core installation. They are: 

• Locally using a command prompt. 

• Remotely, using Remote Desktop. The shell in Remote Desktop will have only the 

same functionality (a command prompt) as being logged on locally. 

• Remotely using Windows Remote Shell. 

• Remotely using an MMC snap-in from a server running Windows Vista or Windows 

Server 2008.  

Some tasks are a bit tricky in Server Core—we’re used to usually doing them exclusively 
from the GUI. An obvious task is changing the password on your account. For that, use the 
net user <username> * command. Some of the tasks that can be a problem. Table 9-2 
shows some solutions.  

Table 9-2 Common Task Workarounds in Server Core 

Task Solution/Workaround 

Enable automatic updates 

 

Cscript %windir%\system32\scregedit.wsf /AU 
[value] 

Where values are: 

1 – disable automatic updates 

4 – enable automatic updates 

/v – view current setting 

Enable Remote Desktop for Administrators Cscript %windir%\system32\scregedit.wsf /AR 
[value] 

Where values are: 

0 – enable Remote Desktop 

1 – disable Remote Desktop 

/v – view current setting 

Enable Terminal Server clients from 

Windows versions prior to Windows Vista 
Cscript %windir%\system32\scregedit.wsf /CS 
[value] 

Where values are: 

0 – enable prior versions 

1 – disable prior versions 

/v – view current setting 

Allow IPSec Monitor remote management Cscript %windir%\system32\scregedit.wsf /IM 
[value] 

Where values are: 

0 – disable remote management 

1 – enable remote management 

/v – view current setting 

Configure DNS SRV record weight and 

priority 
Cscript %windir%\system32\scregedit.wsf /DP 
[value] 

Where DNS SRV priority values are: 

0-65535. (Recommended value = 200) 

/v – view current setting 

Cscript %windir%\system32\scregedit.wsf /DW 
[value] 

Where DNS SRV weight values are: 

0-65535. (Recommended value = 50) 

/v – view current setting 

Update User passwords Net user <username> [/domain] * 

Installing .msi files Use the /q or /qb switches from the command line with the full 

.msi filename. /q is quiet; /qb is quiet but with a basic user 

interface 

Changing the time zone, date, or time timedate.cpl 

Change internationalization settings intl.cpl 

Using Disk Management console From the command line of the Server Core installation: 

Net start VDS 

Then run Disk Management remotely.  

Get Windows version information Winver is not available. Use systeminfo.exe instead. 

Get Help (regular Windows Help and 

Support files are not viewable in Server 

Core) 

Cscript %windir%\system32\scregedit.wsf /cli 

 

Using Windows Remote Shell 

You can use Windows Remote Shell to remotely execute commands on a Server Core 
computer. But before you can run Windows Remote Shell, you need to first enable it on 
the target Server Core computer. To enable Windows Remote Shell, use the following 
command: 

WinRM quickconfig 

To run a command remotely, use the WinRS command from another computer using the 
following command: 

Winrs –r:<ServerName> <command string to execute> 

Using Terminal Server RemoteApp 

One neat trick that we like is to use the new TS RemoteApp functionality of Windows 
Server 2008 to publish a Command Prompt window for the Server Core computer directly 

onto our desktop. This is simpler and more direct, and saves screen real estate, which is 
always a benefit. To create an RDP package that you can put on your desktop, follow 
these steps: 

1. On a Windows Server 2008 server that has the Terminal Services role enabled, open 
the TS RemoteApp Manager, as shown in Figure 9-5.  

 
Figure 9-5 Use the TS RemoteApp Manager to create a remote cmd.exe window. 

2. Connect to the Server Core computer you want to build an RDP package for. 

3. Click Add RemoteApp Programs in the actions pane to open the RemoteApp 
Wizard.  

4. Click Next to open the Choose Programs To Add To The RemoteApp Programs List 
page, shown in Figure 9-6.  

 
Figure 9-6 The Choose Programs To Add To The RemoteApp Programs List page of the 
RemoteApp Wizard. 

5. Click Browse, and navigate to \<ServerName>\c$\windows\system32\cmd.exe. Click 
Open.  

6. Click Next and then click Finish to add the remote program and return to the TS 
RemoteApp Manager.  

7. Select cmd.exe in the RemoteApp programs pane and click Create .rdp File in the 
actions pane.  

8. Click Next, and specify any additional package settings for the RDP package. Note 
the location where the package will be saved. 

9. Click Next and then click Finish to create the RDP package. 

10. Copy the package to the computer where you will use it.  

Now you can open a Command Prompt window directly onto the Server Core computer 
simply by double-clicking the RDP package you created and saved.  

Summary 
In this chapter we’ve covered some basic steps for setting up and configuring the new 
Server Core installation option of Windows Server 2008. We think this is an exciting and 
powerful new way to get the power of Windows Server while maintaining very high levels 
of security and ease of management. And yes, we know that sounds a bit like marketing 
hype, but we actually think that Server Core is an important step forward.  

In the next chapter, we’ll cover managing and configuring your printers using the Printer 
Management console.  

